NAG Library Routine Document

F07HUF (ZPBCON)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07HUF (ZPBCON) estimates the condition number of a complex Hermitian positive definite band matrix A, where A has been factorized by F07HRF (ZPBTRF).

2 Specification

```
SUBROUTINE F07HUF (UPLO, N, KD, AB, LDAB, ANORM, RCOND, WORK, RWORK, INFO)

INTEGER                 N, KD, LDAB, INFO
REAL (KIND=nag_wp)      ANORM, RCOND, RWORK(N)
COMPLEX (KIND=nag_wp) AB(LDAB,*), WORK(2*N)
CHARACTER(1)            UPLO
```

The routine may be called by its LAPACK name ***zpbcon***.

3 Description

F07HUF (ZPBCON) estimates the condition number (in the 1-norm) of a complex Hermitian positive definite band matrix A:

$$\kappa_1(A) = \|A\|_1 \left\|A^{-1}\right\|_1.$$

Since A is Hermitian, $\kappa_1(A) = \kappa_\infty(A) = \|A\|_\infty \left\|A^{-1}\right\|_\infty$.

Because $\kappa_1(A)$ is infinite if A is singular, the routine actually returns an estimate of the **reciprocal** of $\kappa_1(A)$.

The routine should be preceded by a call to F06UEF to compute $\|A\|_1$ and a call to F07HRF (ZPBTRF) to compute the Cholesky factorization of A. The routine then uses Higham's implementation of Hager's method (see Higham (1988)) to estimate $\left\|A^{-1}\right\|_1$.

4 References

Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

5 Parameters

1: UPLO – CHARACTER(1) *Input*

On entry: specifies how A has been factorized.

UPLO = 'U'
$A = U^{\mathrm{H}}U$, where U is upper triangular.

UPLO = 'L'
$A = LL^{\mathrm{H}}$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: KD – INTEGER *Input*

On entry: k_d, the number of superdiagonals or subdiagonals of the matrix A.

Constraint: KD ≥ 0.

4: AB(LDAB,$*$) – COMPLEX (KIND=nag_wp) array *Input*

Note: the second dimension of the array AB must be at least $\max(1, \mathrm{N})$.

On entry: the Cholesky factor of A, as returned by F07HRF (ZPBTRF).

5: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07HUF (ZPBCON) is called.

Constraint: LDAB $\geq$ KD $+ 1$.

6: ANORM – REAL (KIND=nag_wp) *Input*

On entry: the 1-norm of the **original** matrix A, which may be computed by calling F06UEF with its parameter NORM = '1'. ANORM must be computed either **before** calling F07HRF (ZPBTRF) or else from a **copy** of the original matrix A.

Constraint: ANORM ≥ 0.0.

7: RCOND – REAL (KIND=nag_wp) *Output*

On exit: an estimate of the reciprocal of the condition number of A. RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than ***machine precision***, A is singular to working precision.

8: WORK($2 \times$ N) – COMPLEX (KIND=nag_wp) array *Workspace*

9: RWORK(N) – REAL (KIND=nag_wp) array *Workspace*

10: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed estimate RCOND is never less than the true value ρ, and in practice is nearly always less than 10ρ, although examples can be constructed where RCOND is much larger.

8 Further Comments

A call to F07HUF (ZPBCON) involves solving a number of systems of linear equations of the form $Ax = b$; the number is usually 5 and never more than 11. Each solution involves approximately $16nk$ real floating point operations (assuming $n \gg k$) but takes considerably longer than a call to F07HSF (ZPBTRS) with one right-hand side, because extra care is taken to avoid overflow when A is approximately singular.

The real analogue of this routine is F07HGF (DPBCON).

9 Example

This example estimates the condition number in the 1-norm (or ∞-norm) of the matrix A, where

$$A = \begin{pmatrix} 9.39 + 0.00i & 1.08 - 1.73i & 0.00 + 0.00i & 0.00 + 0.00i \\ 1.08 + 1.73i & 1.69 + 0.00i & -0.04 + 0.29i & 0.00 + 0.00i \\ 0.00 + 0.00i & -0.04 - 0.29i & 2.65 + 0.00i & -0.33 + 2.24i \\ 0.00 + 0.00i & 0.00 + 0.00i & -0.33 - 2.24i & 2.17 + 0.00i \end{pmatrix}.$$

Here A is Hermitian positive definite, and is treated as a band matrix, which must first be factorized by F07HRF (ZPBTRF). The true condition number in the 1-norm is 153.45.

9.1 Program Text

```
    Program f07hufe

!     F07HUF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, x02ajf, zlanhb => f06uef, zpbcon, zpbtrf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter                :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)                :: anorm, rcond
      Integer                           :: i, info, j, kd, ldab, n
      Character (1)                     :: uplo
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ab(:,:), work(:)
      Real (Kind=nag_wp), Allocatable  :: rwork(:)
!     .. Intrinsic Procedures ..
      Intrinsic                         :: max, min
!     .. Executable Statements ..
      Write (nout,*) 'F07HUF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kd
      ldab = kd + 1
      Allocate (ab(ldab,n),work(2*n),rwork(n))

!     Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Do i = 1, n
          Read (nin,*)(ab(kd+1+i-j,j),j=i,min(n,i+kd))
        End Do
      Else If (uplo=='L') Then
        Do i = 1, n
          Read (nin,*)(ab(1+i-j,j),j=max(1,i-kd),i)
        End Do
      End If

!     Compute norm of A
!     f06uef is the NAG name equivalent of the LAPACK auxiliary zlanhb
```

```
      anorm = zlanhb('1-norm',uplo,n,kd,ab,ldab,rwork)

!     Factorize A
!     The NAG name equivalent of zpbtrf is f07hrf
      Call zpbtrf(uplo,n,kd,ab,ldab,info)

      Write (nout,*)
      If (info==0) Then

!       Estimate condition number
!       The NAG name equivalent of zpbcon is f07huf
        Call zpbcon(uplo,n,kd,ab,ldab,anorm,rcond,work,rwork,info)

        If (rcond>=x02ajf()) Then
          Write (nout,99999) 'Estimate of condition number =', &
            1.0_nag_wp/rcond
        Else
          Write (nout,*) 'A is singular to working precision'
        End If
      Else
        Write (nout,*) 'A is not positive definite'
      End If

99999 Format (1X,A,1P,E10.2)
    End Program f07hufe
```

9.2 Program Data

```
F07HUF Example Program Data
  4  1                                                    :Values of N and KD
  'L'                                                     :Value of UPLO
 ( 9.39, 0.00)
 ( 1.08, 1.73) ( 1.69, 0.00)
               (-0.04,-0.29) ( 2.65, 0.00)
                             (-0.33,-2.24) ( 2.17, 0.00)  :End of matrix A
```

9.3 Program Results

```
 F07HUF Example Program Results

 Estimate of condition number =  1.32E+02
```